
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2013-03_D14wk49’.

3. Corrections required for TC 8.1.2.8
Change 1 – Correction to local function ‘fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6’
	Function name
	fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6 ()

	Reason for change
	The UE receives during the second PDN establishment, the same IP address and/or Ipv6 Interface ID as it already got in the preamble. This is incorrect: different PDNs shall have different IP addresses.

	Summary of change
	A different IP address and/or Ipv6 Interface ID is assigned during establishment of the second PDN in step 4 and 5 of the local function.  

	TTCN module
	LTE\8_1\RRC_ConnEst.ttcn

	MCC160 Comment
	


Before change

...

  /*
   * @desc      Activate a default DRB connected to an additional DRB
   *            see 36.508 clause 6.4.3.2 steps 2a2 to 6
   * @param     p_CellId
   * @param     p_EpsBearerId2                          .. Identity of the 2nd Default Bearer that will be configured to connect to the additional PDN
   * @param     p_RRCState
   * @status    APPROVED (LTE)
   */
  function fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6(EUTRA_CellId_Type p_CellId,
                                                         EPS_BearerIdentity p_EpsBearerId2,
                                                         EUTRA_RRC_STATE_Type p_RRCState) runs on EUTRA_PTC
  {
    var ProcedureTransactionIdentifier v_PTId_UE;
    var template (omit) ProtocolConfigOptions v_Pco;
    var template (value) AccessPointName v_AccessPointName;
    var PDN_Address v_PDN_Address2;
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var NasCount_Type v_NasCountUL;
    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default
    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);
    // The UE transmits a SERVICE REQUEST
    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.
    if ( p_RRCState == RRC_IDLE ){
      // UE is in RRC Idle mode
      // Steps 2a2 to 2a7
      f_EUTRA_RRC_ConnectionSetup_Def ( eutra_Cell1 );
      v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@
                                                         cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                           cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));
      v_NasCountUL := v_NasInd.SecurityProtection.NasCount;
      // activate security in AS with parameters from AKA performed in preamble
      v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);
      v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(p_CellId,
                                                       v_SecurityParams,
                                                       v_NasCountUL);
      f_EUTRA_Security_Set(v_SecurityParams);
      SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,
                                   cs_TimingInfo_Now,
                                   cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,
                                                                          omit,
                                                                          omit,
                                                                          cs_508_RadioResourceConfigDedicated_SRB2( v_DRB_ToAddModList,
                                                                                                                    cs_508_MAC_MainConfig_Explicit_RBC_DrxL,
                                                                                                                    cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                                                                                               v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                                                                                               v_AntennaInfo)),
                                                                          omit)));
      // receive RRCConnectionReconfigurationComplete
      SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,
                                      cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    } // end of if RRC_Idle
    // UE is now in RRC connected mode
    // Step 3. The UE transmits an ULInformationTransfer message including the PDN CONNECTIVITY REQUEST message.
    SRB.receive ( car_SRB2_NasPdu_IND ( p_CellId,
                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,
                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;
    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;
    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;
    v_AccessPointName := cs_AccessPointName(v_APN_Value);
    v_Pco := v_PDNConnReq.protocolConfigurationOptions;
    // Change from DRB1 to DRB2 to handle U Plane data
    f_EUTRA_StartIPfor2ndPDN(p_CellId); //@sic R5-113734 change 8 sic@
    // Step 4 and 5. The SS transmits an RRCConnectionReconfiguration message including
    // the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message to establish the
    // default EPS bearer context.
    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco));
    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig(p_CellId,
                                                          tsc_RRC_TI_Def,
                                                          p_EpsBearerId2, // Default bearer ID to additional PDN
                                                          v_PTId_UE,  // use value as defined by the UE
                                                          v_AccessPointName,
                                                          v_PDN_Address2,
                                                          f_GetDefaultProtocolConfigOptions(v_Pco));
    // Step 6. The UE transmits an ULInformationTransfer message including the ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message
    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId2))));
    //1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling
    f_EUTRA_DelayForUserPlaneSignalling();
    f_EUTRA_RRC_ConnectionRelease(p_CellId);  // @sic R5s100326 sic@
  } // fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6
...

After change

...

  /*
   * @desc      Activate a default DRB connected to an additional DRB
   *            see 36.508 clause 6.4.3.2 steps 2a2 to 6
   * @param     p_CellId
   * @param     p_EpsBearerId2                          .. Identity of the 2nd Default Bearer that will be configured to connect to the additional PDN
   * @param     p_RRCState
   * @status    APPROVED (LTE)
   */
  function fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6(EUTRA_CellId_Type p_CellId,
                                                         EPS_BearerIdentity p_EpsBearerId2,
                                                         EUTRA_RRC_STATE_Type p_RRCState) runs on EUTRA_PTC
  {
    var ProcedureTransactionIdentifier v_PTId_UE;
    var template (omit) ProtocolConfigOptions v_Pco;
    var template (value) AccessPointName v_AccessPointName;
    var PDN_Address v_PDN_Address2;
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var NasCount_Type v_NasCountUL;
    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default
    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);
    // The UE transmits a SERVICE REQUEST
    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.
    if ( p_RRCState == RRC_IDLE ){
      // UE is in RRC Idle mode
      // Steps 2a2 to 2a7
      f_EUTRA_RRC_ConnectionSetup_Def ( eutra_Cell1 );
      v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@
                                                         cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                           cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));
      v_NasCountUL := v_NasInd.SecurityProtection.NasCount;
      // activate security in AS with parameters from AKA performed in preamble
      v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);
      v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(p_CellId,
                                                       v_SecurityParams,
                                                       v_NasCountUL);
      f_EUTRA_Security_Set(v_SecurityParams);
      SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,
                                   cs_TimingInfo_Now,
                                   cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,
                                                                          omit,
                                                                          omit,
                                                                          cs_508_RadioResourceConfigDedicated_SRB2( v_DRB_ToAddModList,
                                                                                                                    cs_508_MAC_MainConfig_Explicit_RBC_DrxL,
                                                                                                                    cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                                                                                               v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                                                                                               v_AntennaInfo)),
                                                                          omit)));
      // receive RRCConnectionReconfigurationComplete
      SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,
                                      cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    } // end of if RRC_Idle
    // UE is now in RRC connected mode
    // Step 3. The UE transmits an ULInformationTransfer message including the PDN CONNECTIVITY REQUEST message.
    SRB.receive ( car_SRB2_NasPdu_IND ( p_CellId,
                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,
                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;
    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;
    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;
    v_AccessPointName := cs_AccessPointName(v_APN_Value);
    v_Pco := v_PDNConnReq.protocolConfigurationOptions;
    // Change from DRB1 to DRB2 to handle U Plane data
    f_EUTRA_StartIPfor2ndPDN(p_CellId); //@sic R5-113734 change 8 sic@
    // Step 4 and 5. The SS transmits an RRCConnectionReconfiguration message including
    // the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message to establish the
    // default EPS bearer context.
    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2);
    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig(p_CellId,
                                                          tsc_RRC_TI_Def,
                                                          p_EpsBearerId2, // Default bearer ID to additional PDN
                                                          v_PTId_UE,  // use value as defined by the UE
                                                          v_AccessPointName,
                                                          v_PDN_Address2,
                                                          f_GetDefaultProtocolConfigOptions(v_Pco));
    // Step 6. The UE transmits an ULInformationTransfer message including the ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message
    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId2))));
    //1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling
    f_EUTRA_DelayForUserPlaneSignalling();
    f_EUTRA_RRC_ConnectionRelease(p_CellId);  // @sic R5s100326 sic@
  } // fl_EUTRA_ActivateAdditionalPDN_Steps_2a2_to_6
... 
