
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.6.2.12 which is part of the LTE_A test suite based on ‘iwd-TTCN3-B2013-03_D14wk37’ delivery

The test case can be demonstrated to run with one LTE UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_6_2_12
Test Group:
LTE_A_IWD_14wk37
ATS Version:
iwd-TTCN3-B2013-03_D14wk37
System Simulator used:
Anite Conformance Toolset solution
UE’s used:
Samsung Shannon UE

Verification Status:
PASS


4. General correction required for test case 8.6.2.12
Change 1
	Function Name
	f_TC_8_6_2_12_EUTRA

	Reason for change
	1. After step 13 since cell 1 is turned off the SRB’s and DRB’s corresponding to cell 1 should be reset by using function f_EUTRA_SS_SRBs_DRBs_Reset

2. As per 36.331 ASN 1 “The Physical Cell Identity of the PCell the UE was connected to prior to the failure.” At step 14 RRCConnectionReestablishmentRequest, Physical cell Id of cell 1 should be expected rather than cell12
3. PLMN is different in Cell 1 and Cell 12, so UE would perform TAU procdure which is missing in the test procedure on cell 12

4. As “v_TimerValue”is 30s plus tolerance and “p_LoggingInterval_r10” is ‘ms10240’ so UE can do multiple loggings. Hence, “logMeasReport” measurements did on Cell 1 and Cell 12 UE can send multiple reports in “UEInformationReponse” in the list “logMeasInfoList-r10”
5. Wrong cell Id used inside function f_EUTRA_SS_AS_ActivateSecurity_HO_Reest

	Summary of change
	1. SRB and DRB’s are reset for cell 1 using function “f_EUTRA_SS_SRBs_DRBs_Reset”
2. “v_PhysCellId_Cell1” is expected in re-establishment req.
3. TAU procedure is added after step 18
4. Updated TTCN “UEInformationReponse” message to expect multiple values for “logMeasInfoList-r10”
5. Corrected the Cell ID to Cell12 inside function f_EUTRA_SS_AS_ActivateSecurity_HO_Reest
Note: Prose CR R5-145681 is attached with this CR for changes 3 and 4

	Source of change
	RRC_MDT_Logged.ttcn


Before:
	  function f_TC_8_6_2_12_EUTRA() runs on EUTRA_PTC

  { //Logged MDT / Logging and reporting / Reporting at RRC connection re-establishment / PLMN list

    var template (value) CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var Frequency_IE_Type v_Frequency_IE_f2;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_CellInfo_Type v_CellInfo_Cell12;

    var CellIdentity v_CellIdentity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell12;

    var PhysCellId v_PhysCellId_Cell12;

    var float v_TimerValue;

    var B48_Type v_AbsoluteTimeInfo;

    var ShortMAC_I v_ShortMAC_I;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)

    };

    // Init variables

    f_EUTRA_Init(c3);

    v_Frequency_IE_f2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell12);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_CellIdentity_Cell1 := v_CellInfo_Cell1.CellIds.CellIdentity;

    v_PLMN_Identity_Cell1 := v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_CellInfo_Cell12 := f_EUTRA_CellInfo_Get(eutra_Cell12);

    v_PhysCellId_Cell12 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell12);

    v_PLMN_Identity_Cell12 := v_CellInfo_Cell12.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_TimerValue := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 30.0);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell12);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    //The UE is registered on PLMN1 (Cell 1) using the procedure described in TS 36.508[18]

    //clause 4.5.2.3 except that the ATTACH ACCEPT message indicates PLMN2 in

    //the Equivalent PLMN list as described in Table  8.6.2.12.3.3-6

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);

    //Bring UE to inital state with Equivalent PLMN2

    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell1,

                                   PREAMBLE,

                                   cs_PLMN_List_1PLMN (f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell12))));

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits a LoggedMeasurementConfiguration message on Cell 1.

    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                 cs_TimingInfo_Now,

                                 cs_LoggedMeasurementConfiguration_with_v1080_IEs(v_PLMN_Identity_Cell1, '0000EF'O, '001A'O, '05'O, v_AbsoluteTimeInfo, -, -, -,

                                                                                  cs_LoggedMeasurementConfiguration_v1080_IEs({v_PLMN_Identity_Cell1, v_PLMN_Identity_Cell12}))));

    //@siclog "Step 2" siclog@

    //The SS transmits an RRCConnectionRelease message to release the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    //@siclog "Step 3" siclog@

    //Wait 30 seconds for UE performing the logging at regular time intervals

    f_Delay(v_TimerValue);

    //@siclog "Step 4 - 12" siclog@

    //Steps 1 to 9 of the generic radio bearer establishment procedure in TS 36.508 subclause 4.5.3.3 are executed

    //to successfully complete the service request procedure on Cell 1.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    //@siclog "Step 13" siclog@

    //The SS changes Cell 1 and Cell 12 parameters according to the row "T1" in table 8.6.2.12.3.2-1 in order that

    //the radio link quality of Cell 1 is degraded and Cell 12 is suitable for camping.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE in Step 2

    v_ShortMAC_I := f_Calculate_ShortMAC(eutra_Cell1, eutra_Cell12);

    //@siclog "Step 14" siclog@

    //The UE sends RRCConnectionReestablishmentRequest message on Cell 12.

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell12,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysCellId_Cell12,
                                                                               cr_ReestablishmentCause_OtherFailure,

                                                                               v_ShortMAC_I)));

    v_Auth_Params := f_EUTRA_Security_Get ( );

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell2, v_Auth_Params, 0);//Change 5
    f_EUTRA_Security_Set ( v_Auth_Params );

    //@siclog "Step 15" siclog@

    //The SS transmits RRCConnectionReestablishment message.

    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell12 );

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit RRCConnectionReestablishmentComplete message with logMeasAvailable set to true?

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell12,

                                    cr_RRCConnectionReestablishmentComplete_r10(tsc_RRC_TI_Def, true_, true_)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.12 Step 16");

    //@siclog "Step 17 - 18" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to resume the existing radio bearer.

    //The UE transmits an RRCConnectionReconfigurationComplete message.

    f_EUTRA_RRC_RRCConnectionReconfiguration_Resume_SRB2_DRBs(eutra_Cell12, tsc_RRC_TI_Def);

    //@siclog "Step 19" siclog@

    //The SS sends a UEInformationRequest message with logMeasReportReq present.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell12,

                                 cs_TimingInfo_Now,

                                 cs_508_UEInformationRequest(tsc_RRC_TI_Def)));

    //@siclog "Step 20" siclog@

    //Check: Does the UE send a UEInformationResponse message including logMeasReport?

    SRB.receive(car_SRB2_RrcPdu_IND(eutra_Cell12,

                                    cr_UEInformationResponse_LogMeasInfo_1Entry(tsc_RRC_TI_Def, v_AbsoluteTimeInfo,

                                                                                v_PLMN_Identity_Cell1,'0000EF'O, '001A'O, '05'O, *,

                                                                                cr_Cgi(v_PLMN_Identity_Cell1, v_CellIdentity_Cell1),

                                                                                cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f2.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                         v_PhysCellId_Cell12,

                                                                                                                         omit))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.12 Step 20");

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell12, E2_CONNECTED);

  } // end of f_TC_8_6_2_12_EUTRA


After:
	  function f_TC_8_6_2_12_EUTRA() runs on EUTRA_PTC

  { //Logged MDT / Logging and reporting / Reporting at RRC connection re-establishment / PLMN list

    var template (value) CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var Frequency_IE_Type v_Frequency_IE_f2;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_CellInfo_Type v_CellInfo_Cell12;

    var CellIdentity v_CellIdentity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell1;

    var PLMN_Identity v_PLMN_Identity_Cell12;

    var PhysCellId v_PhysCellId_Cell12;

    var PhysCellId v_PhysCellId_Cell1;//Change 2
    var float v_TimerValue;

    var B48_Type v_AbsoluteTimeInfo;

    var ShortMAC_I v_ShortMAC_I;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var SRB_COMMON_IND v_ReceivedAsp; //Anite changes

    var LogMeasInfo_r10 v_LogMeasInfo_r10; //Anite changes

    var integer v_FieldCnt; //Anite changes

    var integer i;  //Anite changes
    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)

    };

    // Init variables

    f_EUTRA_Init(c3);

    v_Frequency_IE_f2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell12);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1); //Change 2
    v_CellIdentity_Cell1 := v_CellInfo_Cell1.CellIds.CellIdentity;

    v_PLMN_Identity_Cell1 := v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_CellInfo_Cell12 := f_EUTRA_CellInfo_Get(eutra_Cell12);

    v_PhysCellId_Cell12 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell12);

    v_PLMN_Identity_Cell12 := v_CellInfo_Cell12.NAS_Parameters.Guti_Parameters.PLMN_Identity;

    v_TimerValue := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 30.0);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell12);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    //The UE is registered on PLMN1 (Cell 1) using the procedure described in TS 36.508[18]

    //clause 4.5.2.3 except that the ATTACH ACCEPT message indicates PLMN2 in

    //the Equivalent PLMN list as described in Table  8.6.2.12.3.3-6

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);

    //Bring UE to inital state with Equivalent PLMN2

    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell1,

                                   PREAMBLE,

                                   cs_PLMN_List_1PLMN (f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell12))));

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits a LoggedMeasurementConfiguration message on Cell 1.

    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                 cs_TimingInfo_Now,

                                 cs_LoggedMeasurementConfiguration_with_v1080_IEs(v_PLMN_Identity_Cell1, '0000EF'O, '001A'O, '05'O, v_AbsoluteTimeInfo, -, -, -,

                                                                                  cs_LoggedMeasurementConfiguration_v1080_IEs({v_PLMN_Identity_Cell1, v_PLMN_Identity_Cell12}))));

    //@siclog "Step 2" siclog@

    //The SS transmits an RRCConnectionRelease message to release the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    //@siclog "Step 3" siclog@

    //Wait 30 seconds for UE performing the logging at regular time intervals

    f_Delay(v_TimerValue);

    //@siclog "Step 4 - 12" siclog@

    //Steps 1 to 9 of the generic radio bearer establishment procedure in TS 36.508 subclause 4.5.3.3 are executed

    //to successfully complete the service request procedure on Cell 1.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    //@siclog "Step 13" siclog@

    //The SS changes Cell 1 and Cell 12 parameters according to the row "T1" in table 8.6.2.12.3.2-1 in order that

    //the radio link quality of Cell 1 is degraded and Cell 12 is suitable for camping.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE in Step 2

    v_ShortMAC_I := f_Calculate_ShortMAC(eutra_Cell1, eutra_Cell12);

    f_EUTRA_SS_SRBs_DRBs_Reset (eutra_Cell1); //Change 1
    //@siclog "Step 14" siclog@

    //The UE sends RRCConnectionReestablishmentRequest message on Cell 12.

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell12,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysCellId_Cell1,//Change 2
                                                                               cr_ReestablishmentCause_OtherFailure,

                                                                               v_ShortMAC_I)));

    v_Auth_Params := f_EUTRA_Security_Get ( );

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell12, v_Auth_Params, 0);//Change 5
    f_EUTRA_Security_Set ( v_Auth_Params );

    //@siclog "Step 15" siclog@

    //The SS transmits RRCConnectionReestablishment message.

    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell12 );

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit RRCConnectionReestablishmentComplete message with logMeasAvailable set to true?

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell12,

                                    cr_RRCConnectionReestablishmentComplete_r10(tsc_RRC_TI_Def, true_, true_)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.12 Step 16");

    //@siclog "Step 17 - 18" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to resume the existing radio bearer.

    //The UE transmits an RRCConnectionReconfigurationComplete message.

    f_EUTRA_RRC_RRCConnectionReconfiguration_Resume_SRB2_DRBs(eutra_Cell12, tsc_RRC_TI_Def);

    f_EUTRA_ConnectedMode_TAU(eutra_Cell12); //Change 3
    //@siclog "Step 19" siclog@

    //The SS sends a UEInformationRequest message with logMeasReportReq present.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell12,

                                 cs_TimingInfo_Now,

                                 cs_508_UEInformationRequest(tsc_RRC_TI_Def)));

    //@siclog "Step 20" siclog@

    //Check: Does the UE send a UEInformationResponse message including logMeasReport?

    //change 4

    SRB.receive(car_SRB2_RrcPdu_IND(eutra_Cell12,

                                    cr_508_UEInformationResponse(tsc_RRC_TI_Def, omit, omit, cr_LogMeasReport_r10(v_AbsoluteTimeInfo, v_PLMN_Identity_Cell1,'0000EF'O, '001A'O, '05'O, *, *))))

      -> value v_ReceivedAsp; //@sic R5s130525 sic@

    v_FieldCnt := lengthof(v_ReceivedAsp.Signalling.Rrc.Dcch.message_.c1.ueInformationResponse_r9.criticalExtensions.c1.ueInformationResponse_r9.nonCriticalExtension.nonCriticalExtension.logMeasReport_r10.logMeasInfoList_r10); //@sic R5s130525 sic@

    if (v_FieldCnt < 2) {f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 20: Not enough LogMeasurementInfo entries")} //@sic R5s130525 sic@

    for (i:=0; i<v_FieldCnt; i:=i+1) {

      v_LogMeasInfo_r10 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.c1.ueInformationResponse_r9.criticalExtensions.c1.ueInformationResponse_r9.nonCriticalExtension.nonCriticalExtension.logMeasReport_r10.logMeasInfoList_r10[i];

      if (not match(v_LogMeasInfo_r10, cr_LogMeasInfo_r10(*, ?, cr_Cgi(v_PLMN_Identity_Cell1,v_CellIdentity_Cell1), (0..97), (0..34), cr_MeasResultNeighCells_r10_EUTRA_1Entry(v_Frequency_IE_f2.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                                         v_PhysCellId_Cell12,

                                                                                                                         omit))))

        {

          if (i == v_FieldCnt - 1) {f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21: None of the LogMeasurementInfo matched the template")}

        }

      else

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step21"); break;

        }

    } 

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.12 Step 20");

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell12, E2_CONNECTED);

  } // end of f_TC_8_6_2_12_EUTRA


5. Execution Log Files

1) Samsung Shannon UE
The Samsung Shannon UE passed this test case on Anite Conformance Toolset Solution in LTE FDD path on band 3. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_6_2_12_UE1_Log.htm   

	[1]
	R5s141293:    Supporting information for agreement of LTE-A Rel-11 eMDT Test Case 8.6.2.12. This archive comprises: html and text format execution log files as well as PICS/PIXIT settings.

	
	


